

ФЕДЕРАЛЬНОЕ АГЕНТСТВО ПО ОБРАЗОВАНИЮ
ВОРОНЕЖСКИЙ ГОСУДАРСТВЕННЫЙ УНИВЕРСИТЕТ

Язык программирования Pascal
Множества. Типизированные константы

Практикум

Специальность 010101 (010100) Математика

ВОРОНЕЖ
2005

Утверждено научно-методическим советом Математического факультета

— (28 февраля 2005 года, протокол №6)

Составители: Васильев В.В., Хливненко Л.В.

Практикум подготовлен на кафедре математического моделирования математического факультета Воронежского государственного университета.

Рекомендуется для студентов вечернего отделения математического факультета Воронежского государственного университета.

Множества относятся к структурированным типам данных и представляют собой наборы элементов одного типа. Порядок элементов при описании не учитывается. Элементы не нумеруются.

Множества используются в задачах, для которых важен факт наличия или отсутствия элемента во множестве себе подобных.

В Паскале множества могут состоять только из элементов порядковых типов. Все элементы конкретного множества должны принадлежать одному типу, который называется **базовым типом множества**. Максимальное количество значений базового типа называется мощностью множества. В качестве базового типа множества в Паскале допустимы только типы, мощность которых не больше 256. Кроме того, номера нижней и верхней границ базового типа должны находиться в диапазоне 0..255.

Поэтому базовыми типами не могут быть типы Integer, LongInt, Word, Real. Для различимости элементы множества должны быть представлены уникальными значениями. Множество, не содержащее элементов, называется **пустым** и обозначается [].

Множественный тип в общем виде задается следующим образом:

Set of <Базовый тип множества>;

Например, тип множества, состоящего из цифр, можно описать так:

Type Set of 0..9;

Сравнение множеств. Два множества равны, если все их элементы одинаковы, причем порядок следования элементов не имеет значения. Например, [1,2,3]=[3,1,2].

Результатом операции сравнения $A \leq B$ ($A \subseteq B$) будет true, если множество **B** включает все элементы множества **A** (**A** является подмножеством **B**).

Результатом операции сравнения $A \geq B$ ($A \supseteq B$) будет true, если множество **A** включает все элементы множества **B** (**B** является подмножеством **A**).

Отношение строгого включения ($A \subset B$) можно сделать следующим образом: $(A \geq B) \text{ And } (A <> B)$.

Операции над множествами. В Паскале над множествами допустимы операции объединения, пересечения и разности.

Результатом операции **сложения** (*объединения*) множеств **A** и **B** является множество **A+B**, содержащее элементы множеств **A** и **B** (*без дублирования одинаковых элементов*). Все элементы множества уникальны!

Результатом операции **вычитания** (*разности*) множеств **A** и **B** является множество **A - B**, содержащее элементы множества **A**, не входящие во множество **B**.

Результатом операции **умножения** (*пересечения*) множеств **A** и **B** является множество **A * B**, содержащее элементы, входящие и во множество **A**, и во множество **B**.

Проверить, **принадлежит** ли элемент множеству, можно с помощью знаковой нам логической операции IN: Элемент **IN** Множество

Конструктором множеств называется выражение множественного ти-

па, которое может содержать операции объединения, разности и пересечения множеств. Конструктор множеств может стоять в правой части оператора присваивания.

❗ Для того, чтобы добавить или удалить элемент, его представляют как множество мощности 1. Для этого элемент берут в квадратные скобки !

Для добавления нового элемента во множество имеется специальная процедура INCLUDE (Множество, Элемент);

Для исключения элемента из множества служит специальная процедура EXCLUDE (Множество, Элемент);

Используя множества, можно упростить решение задач обработки текстовой информации.

Задача 1. Напишите программу, которая проверяет, имеются ли в слове повторяющиеся буквы.

♣ Словом будем считать последовательность символов, оканчивающуюся точкой. Буквы слова будем записывать в массив из символов. Параллельно с вводом букв будем формировать множество букв слова. Если очередная введенная буква не входит в сформированное на текущий момент множество букв, то буква включается во множество букв слова. В противном случае логической переменной flag (хранящей ответ на вопрос задачи) присваивается значение true. ♠

```

Program Repeat_Letters;
Uses crt;
Const Max_Len = 100; {максимальная длина слова}
        eot = '.'; {признак конца ввода}
Type Word = array [1..Max_Len] of char; {тип слова}
Var Sl: Word; {слово}
        Letters: set of char; {множество букв слова}
        i: 0..Max_Len; {счетчик букв в слове}
        flag: Boolean; {признак наличия повторяющихся букв}
Begin
    Textbackground(7); Textcolor(blue); Clrscr;
    writeln('Введите слово (в конце поставьте точку): ');
    i:=0; Letters:=[]; flag:=false;
    repeat
        inc(i); read(Sl[i]);
        if not (Sl[i] in Letters)
            then Letters:=Letters + [Sl[i]]
            else if not flag
                then flag:=true;
    until (i=Max_Len) or (Sl[i]=eot);
    if flag
        then writeln('В слове есть повторяющиеся буквы! ');
        else writeln('В слове нет повторяющихся букв! ');
    readkey
End. {Repeat_Letters}

```

Научимся проверять, входят ли элементы одного множества в другое.

Задача 2. Пусть дан текст. Верно ли, что в нем имеются буквы, входящие в слово, задаваемое пользователем?

♣ Словом и текстом будем считать последовательность символов, оканчивающуюся точкой. Параллельно с вводом слова сформируем множество его букв. Буква слова включается во множество букв, если она встречается в слове впервые. После завершения ввода слова из множества букв слова исключается точка. *Почему?*

Параллельно с вводом символов текста, проверяется наличие вводимых символов во множестве букв слова. Если проверка даст положительный результат, то логической переменной *flag* (*хранящей ответ на вопрос задачи*) присваивается значение *true*. ♠

```

Program Words1;
Uses crt;
Const Max_Len = 100; {максимальная длина слов}
        eot = '.'; {признак конца ввода}
Type Word = array [1..Max_Len] of char; {тип слова}
Var Sl,Text: Word; {слово и текст}
        Letters: set of char; {множество букв слова}
        i: 0..Max_Len; {счетчик букв}
        flag: Boolean; {признак наличия букв слова в тексте}
Begin
    Textbackground(7); Textcolor(blue); Clrscr;
    {Ввод слова и инициализация множества букв слова}
    writeln('Введите слово (в конце поставьте точку): ');
    i:=0; Letters:=[];
    repeat
        inc(i); read(Sl[i]);
        if Letters-[Sl[i]]=Letters
            then Letters:=Letters+[Sl[i]]
    until (i=Max_Len) or (Sl[i]=eot);
    Letters:=Letters-[eot];
    {Ввод текста и проверка вхождения букв слова в текст}
    writeln('Введите текст (в конце поставьте точку): ');
    i:=0; flag:=false;
    repeat
        inc(i); read(Text[i]);
        if (not flag) and (Text[i] in Letters)
            then flag:=true;
    until (i=Max_Len) or (Text[i]=eot);
    if flag then writeln('В тексте есть буквы слова! ');
    else writeln('В тексте нет букв слова! ');
    readkey
End.{Words1}

```

♣ Приведем второй способ решения данной задачи. Параллельно с вводом

слова и текста формируется множество букв слова и множество букв текста. Если пересечение двух сформированных множеств непусто, то на вопрос задачи надо дать положительный ответ. ♠

```

Program Words;
Uses crt;
Const Max_Len = 100; {максимальная длина слов}
        eot = '.'; {признак конца ввода}
Type Word = array [1..Max_Len] of char; {тип слова}
Var Sl,Text: Word; {слово и текст}
        Letters1, Letters2: set of char; {множество букв слова}
        i: 0..Max_Len; {счетчик букв}
        flag: Boolean; {признак наличия букв слова в тексте}

Begin
    Textbackground(7); Textcolor(blue); Clrscr;
    {Ввод слова и инициализация множества букв слова}
    writeln('Введите слово (в конце поставьте точку): ');
    i:=0; Letters1:=[];
    repeat
        inc(i); read(Sl[i]);
        if Letters1-[Sl[i]]=Letters1
            then Letters1:=Letters1+[Sl[i]]
    until (i=Max_Len) or (Sl[i]=eot);
    Letters1:=Letters1-[eot];
    {Ввод текста и инициализация множества букв текста}
    writeln('Введите текст (в конце поставьте точку): ');
    i:=0; Letters2:=[];
    repeat
        inc(i); read(Text[i]);
        if Letters2-[Text[i]]=Letters2
            then Letters2:=Letters2+[Text[i]]
    until (i=Max_Len) or (Text[i]=eot);
    Letters2:=Letters2-[eot];
    if Letters1*Letters2<>[]
        then writeln('В тексте есть буквы слова! ')
        else writeln('В тексте нет букв слова! ');
    readkey
End. {Repeat_Letters}

```

Задача 3. Пусть дана непустая последовательность слов из строчных русских букв. Между соседними словами присутствует запятая, за последним – точка. Напечатайте в алфавитном порядке гласные буквы, которые входят в каждое слово.

♣ Текст из слов оформим в виде последовательности символов, оканчивающейся точкой. При просмотре очередного слова текста будем формировать множество Gl_Sl его гласных букв. Множество Gl_T гласных букв, входящих во все слова текста, вначале совпадает с множеством гласных букв первого слова.

При просмотре n-го слова текста множество Gl_T получается как результат пересечения множества Gl_T гласных букв, входящих во все слова от 1-го до n-1-го, и множества Gl_Sl гласных букв n-го слова.

Если после просмотра всех слов текста множество Gl_T оказывается не-пусто, то его элементы выводятся в алфавитном порядке. *Обратите внимание на то, как просто вывести элементы множества в алфавитном порядке (см. программу)! ♠*

```

Program Glas;
Uses crt;
Const Max_Len = 100; {максимальная длина текста}
        eot = '.'; {признак конца текста}
        eow = ','; {признак конца слова}
Type TWord = array [1..Max_Len] of char; {тип слова}
        TGl = set of char; {тип множества гласных букв}
Const Gl: TGl =
['a', 'o', 'ы', 'э', 'у', 'я', 'ё', 'и', 'е', 'ю'];
Var Text: TWord; {слово}
      Gl_Sl, Gl_T: set of char;
                {множество гласных букв слова и всех слов}
      i, T_Len: 0..Max_Len; {счетчик букв и длина текста}
      f: boolean; {признак начального заполнения множества Gl_W}
      ch: char; {счетчик цикла}

Begin
  Textbackground(7); Textcolor(blue); Clrscr;
  {Ввод текста}
  writeln('Введите текст (в конце поставьте точку): ');
  i:=0;
  repeat
    inc(i); read(Text[i])
  until (i=Max_Len) or (Text[i]=eot);
  if Text[i]=eot then T_Len:=i-1 else T_Len:=i;
  {Поиск гласных букв, входящих во все слова}
  Gl_T:=[]; f:=true; i:=1;
  repeat
    {пропуск запятых}
    while (i<=T_Len) and (Text[i]=eow) do inc(i);
    {формирование множества гласных букв слова}
    Gl_Sl:=[];
    while (i<=T_Len) and (Text[i]<>eow) do
      begin
        if Text[i] in Gl
          then include(Gl_Sl,Text[i]); inc(i)
      end;
    if f then begin Gl_T:=Gl_Sl; f:=not f end
    else Gl_T:=Gl_T*Gl_Sl

```

```

until (i>=T_Len);
{вывод результата}
if Gl_T=[]
  then writeln('Гласных букв, которые входят в каждое слово, нет!')
  else
    begin
      writeln('Гласные буквы, которые входят в каждое слово:');
      for ch := 'a' to 'я' do
        if ch in Gl_T then write(ch, ' ');
      end;
    end;
readkey
End. {Glas}

```

В программе к задаче 3 нам впервые встретились **структурированные типизированные** (*типированные*) **константы** (см. const Gl).

Иногда в задачах бывает нужно использовать константы структурированного типа. Так, например, в программе Glas использовалось множество Gl гласных строчных русских букв.

В общем виде описать типизированную константу можно так:

Const Имя константы: Тип =(Значение);

Пример константы типа запись:

```

Type TDATA = record gd: 1900..2100; mc: 1..12; dn:1..31
end;

```

```

Const Now: TDATA = (dn: 31; mc: 12; gd: 2004);

```

Пример константы – многомерного массива:

```

Const M: Array [1..2, 1..3, 1..4] of byte =
  (( (1, 2, 3, 4), (5, 6, 7, 8), (9, 10, 11, 12) ),
   ((13, 14, 15, 16), (17, 18, 19, 20), (21, 22, 23, 24) ));

```

Разберем задачу на сообразительность, алгоритм решения которой приведет нас к множествам.

Задача 4. Коту снится, что его окружили тринадцать мышей. Двенадцать из них серые, а одна белая. Слышит кот, что кто-то говорит ему: «Мурлыка, ты можешь съесть каждую тринадцатую мышку. Считай их по кругу в одном направлении. Белую мышку ты должен съесть последней.». Задумался кот: «С какой мышки начинать счет?»

♣ Мыши стоят по кругу. Будем считать, что нумерация мышей в круге начинается с белой мыши и идет по часовой стрелке.

Задачу будем решать перебором всех возможных ситуаций. Начнем счет с первой мыши. Сформируем множество номеров мышей. Будем исключать из этого множества каждую тринадцатую мышку. Если на каком-то этапе при исключении мы попадем в белую мышку, то выбор начала отсчета неудачен, и нужно начать отсчет со следующей мыши. Если белая мышка останется последней, то точка отсчета найдена.

Если номер мыши, только что исключенной из круга, равен i , то следующей будет исключена мышка, номер которой определяется следующим образом:

for i:=1 to 13 do i:=Номер мыши, стоящей в кругу за i;

Когда все мыши стоят в кругу от 1 до max (max=13), то переход от одной мыши к другой выражается оператором $i:=i \bmod \max + 1$. Если при переходе мы попадаем в мышь, которой уже нет в круге, то номер этой мыши пропускается.

```

Program Cat;
Const Max=13; {число мышей}
Var krug:set of 1..max; {множество мышей}
    i,j:byte; {параметры циклов}
    k:1..max; {искомый номер}
Begin
    for k:=1 to max do
        begin
            krug:=[1..max]; i:=k;
            repeat
                for j:=1 to max do
                    repeat
                        i:=i mod max + 1
                    until i in krug;
                    if i<>1 then krug:=krug-[i]
                until (i=1) or (krug=[1]);
                if krug=[1] then break
            end;
            write('Номер мыши, с которой надо начать счет: ',k);
            readln
        end. {Cat}

```

Решим задачу, в которой используется массив из множеств.

Задача 5. Пусть дан фрагмент программы. Найдите значения A, B, C типа ассортимент, перечисленные ниже.

```

type
    продукт = (хлеб, масло, молоко, мясо, рыба, соль, колбаса, сахар, кофе) ;
    ассортимент = set of продукт;
    магазины = array[1..20] of ассортимент;

```

- а) **A** – продукты, которые есть во всех магазинах;
- б) **B** – продукты, которые есть хотя бы в одном магазине;
- в) **C** – продукты, которых нет ни в одном магазине.

♣ Множество **A** получается как пересечение множеств ассортиментов продуктов всех магазинов. Множество **B** есть объединение множеств ассортиментов продуктов всех магазинов. Множество **C** получается как разность множества, состоящего из всех возможных продуктов, и множества **B**. ♠

```

Program Shops_;
Uses crt;
Const Max=20; {количество магазинов}
Type
    products=( bread,butter,milk,meat,fish,salt,cheese,sausage,sugar,coffee ) ;
    assortiment=set of products;

```

```

shops=array[1..max] of assortment;
Var mag:shops; {массив множеств}
    all,one,no:assortiment; {искомые множества}
    n,i:byte; {параметры циклов}
    p:products; {продукт}
Procedure print(m:assortiment);
begin
    for p:=bread to coffee do
        if p in m
            then
                case p of
                    bread:write('хлеб '); butter:write('масло ');
milk:write('молоко ');
                    meat:write('мясо '); fish:write('рыба ');
salt:write('соль ');
                    cheese:write('сыр '); sausage:write('колбаса ');
sugar:write('сахар ');
                    coffee:write('кофе ')
                end;
            writeln
        end; {print}
Begin
    clrscr;
    {ввод ассортимента в магазинах}
    for i:=1 to max do
        begin
            mag[i]:=[];
            writeln('Введите номера продуктов, которые есть в ',i,'-м магазине');
            write('1-хлеб,2-масло,3-молоко,4-мясо,5-рыба,6-
соль,');
            writeln('7-сыр,8-колбаса,9-сахар,10-кофе');
            writeln('0-конец ввода');
            repeat
                repeat
                    read(n)
                until (n>=0) and (n<=10);
                case n of
                    1:p:=bread; 2:p:=butter; 3:p:=milk; 4:p:=meat; 5:p:=fish; 6:p:=salt; 7:p:=cheese;
                    8:p:=sausage; 9:p:=sugar; 10:p:=coffee
                end;
                if not(p in mag[i])
                    then include(mag[i],p)
                until (n=0);
            end;
        {поиск продуктов, которые есть во всех магазинах}
        all:=mag[1];

```

```

for i:=2 to max do all:=all*mag[i];
writeln('Продукты, которые есть во всех магазинах:');
print(all);
{формирование множества продуктов, которые есть хотя бы в одном магазине}
one:=[];
for i:=1 to max do one:=one+mag[i];
writeln('Продукты, которые есть хотя бы в одном магазине:');
print(one);
{формирование множества продуктов, которых нет ни в одном магазине}
no:=[bread..coffee]-one;
writeln('Продукты, которых нет ни в одном магазине:');
print(no);
readkey
End. {Shops_}

```

❗ *Подготовьтесь к ответам на **все(!)** контрольные вопросы и выполните **все(!)** контрольные задания. Дорогу осилит идущий!*

Контрольные вопросы и задания. Множества. Типизированные константы.

1. Что такое множество в Паскале?
2. В каких задачах целесообразнее использовать множества?
3. Дайте определение элемента множества.
4. Что такое базовый тип множества?
5. Каким может быть базовый тип множества?
6. Что в Паскале называют мощностью множества?
7. Какова максимальная мощность множества?
8. Каков диапазон варьирования номеров нижней и верхней границ базового типа?
9. Почему базовым типом множества не могут быть типы Integer, LongInt, Word, Real?
10. Может ли множество содержать несколько одинаковых элементов?
11. Что такое пустое множество? Как оно обозначается?
12. Как в общем виде задается множественный тип?
13. Какие множества считаются равными? Неравными? Учитывается ли при сравнении множеств порядок следования элементов?
14. Как проверить строгое вложение множеств?
15. Что является результатом сложения множеств?
16. Что является результатом вычитания множеств?
17. Что является результатом умножения множеств?
18. Как проверить принадлежность элемента множеству?
19. Что такое конструктор множества?
20. Как добавить (удалить) элемент во (из) множество(а)?
21. Для чего нужны процедуры INCLUDE и EXCLUDE? Каковы аргументы этих процедур?

22. Что такое типизированная константа?
23. Как в общем виде задать типизированную константу?
24. Опишите типизированную константу – множество.
25. Опишите типизированную константу – многомерный массив.
26. Опишите типизированную константу – запись.
27. Напишите программу, которая проверяет, имеются ли в слове повторяющиеся буквы.
28. Пусть дан текст. Верно ли, что в нем имеются буквы, входящие в слово, задаваемое пользователем? Приведите два способа решения задачи.
29. Пусть дана непустая последовательность слов из строчных русских букв. Между соседними словами присутствует запятая, за последним – точка. Напечатайте в алфавитном порядке гласные буквы, которые входят в каждое слово.
30. Коту снится, что его окружили тринадцать мышей. Двенадцать из них серые, а одна белая. Слышит кот, что кто-то говорит ему: «Мурлыка, ты можешь съесть каждую тринадцатую мышку. Считай их по кругу в одном направлении. Белую мышку ты должен съесть последней». Задумался кот: «С какой мышки начинать счет?»
31. Пусть дан фрагмент программы. Найдите значения A, B, C типа ассортимент, перечисленные ниже.

type

продукт = (хлеб, масло, молоко, мясо, рыба, соль, колбаса, сахар, кофе) ;

ассортимент = **set of** продукт ;

магазины = **array**[1..20] **of** ассортимент ;

- а) **A** – продукты, которые есть во всех магазинах;
 - б) **B** – продукты, которые есть хотя бы в одном магазине;
 - в) **C** – продукты, которых нет ни в одном магазине.
32. Пусть дан фрагмент программы.
Var s: set of char; c, d: char;
Присвойте переменной s перечисленное ниже множество литер, которые больше c, но меньше d (c<d).
 33. Вычислите значения отношений:
 - а) [6] <> [6,6,6];
 - б) ['1', '8'] = ['8', '1'];
 - в) ['n', 'm'] = ['n'.. 'm'];
 - г) odd(8) in [];
 - д) trunc[6.7] in (1,5,6).
 34. Вычислите значения выражений:
 - а) [4,6,8]+[5,7];
 - б) [4,6,8]-[5,7];
 - в) [4,6,8]*[5,7].
 35. Вычислите значение выражения (['0' .. '7'] + ['2' .. '9'])*(['a']+['z']).
 36. Упростите (A и B - множества): A*B-A и (A-B)+(B-A)+A*B.
 37. Пусть дан текст. Посчитайте общее число вхождений английских букв в

текст.

38. Пусть вводится последовательность символов длиной не более 4. Если все вводимые символы – цифры, то выполните преобразование данной последовательности в целое число.
39. Пусть дан текст, состоящий из строчных латинских букв и цифр. Определите, каких букв – гласных (а,е,і,о,и) или согласных – больше в этом тексте.
40. Пусть дана непустая последовательность слов из строчных русских букв. Между соседними словами присутствует запятая, а за последним – точка. Напечатайте в алфавитном порядке все звонкие согласные буквы, которые входят более чем в одно слово.
41. Не используя дополнительных переменных, поменяйте местами значения А и В множественного типа.
42. Пусть дан фрагмент программы. Опишите логическую функцию ВЕЗДЕ(ГР), определяющую есть ли в группе ГР хотя бы один человек, побывавший в гостях у всех остальных из группы (ГР[х] – множество людей, побывавших в гостях у человека с именем х).

Type

Имя = (Вика, Света, Миша, Женя, Таня, Лена, Марина, Сергей, Дима, Оля) ;

Гости = **set of** Имя;

Группа = **array** [Имя] **of** Гости;

43. Пусть дано натуральное число n ($n \geq 2$). Найдите все меньшие n простые числа, используя **решето Эратосфена**. Решетом Эратосфена называют следующий способ определения простых чисел. Выпишем подряд все целые числа от 2 до n . Первое простое число 2. Подчеркнем его, а все большие числа, кратные 2, зачеркнем. Первое из оставшихся чисел – 3. Подчеркнем его, а все большие числа, кратные 3, зачеркнем. Первое число из оставшихся теперь – 5, так как 4 уже зачеркнуто. Подчеркнем число 5 как простое, а все большие числа, кратные 5, зачеркнем и т.д.
44. **Индивидуальное(!) задание**, которое передается преподавателю перед началом собеседования по этой теме:

Номер индивидуального задания определяет преподаватель!

Опишите постановку задачи, создайте математическую модель ее решения, разработайте блок-схему и работающую программу, проведите тестирование и отладку программы, обдумайте полученные результаты.

Индивидуальные задания.

1. Пусть задана произвольная последовательность символов. Признак конца последовательности – точка. Напечатайте те символы, которые встречаются в данной последовательности более одного раза.
2. Пусть дан текст, заканчивающийся точкой. Текст состоит из символов, разделенных пробелами. Слово – последовательность латинских букв. Напечатайте слова текста, имеющие нечетный номер, в которых нет ни одной повторяющейся буквы.
3. Пусть задана целочисленная квадратная матрица размерности n . Элементы

- матрицы находятся в диапазоне от 1 до 100. Напечатайте все цифры из заданного диапазона, которых нет ни в одной из строк заданной матрицы.
4. Напечатайте все целые числа, лежащие в диапазоне от 5 до 2500, которые представимы в виде $5n+7m$, где n и m – целые числа ($n, m \geq 0$).
 5. В классе учатся 25 учеников. Каждому ученику были выставлены оценки за четверть по 15 предметам. Определите, сколько в классе отличников, хорошистов и троечников.
 6. Известны результаты анкетирования ста человек. Анкета состоит из 150 пунктов, на которые предлагалось ответить утвердительно или «нет определенного мнения по данному вопросу». Напечатайте номера тех пунктов анкеты, на которые были получены только утвердительные и только отрицательные ответы всех опрошенных (*если, конечно, такие пункты есть*).
 7. В городе N имеется 100 кондитерских магазинов. Известно, что в каждом из этих магазинов не более 20 видов сладостей в ассортименте. Какие виды сладостей есть во всех имеющихся магазинах? Существует ли магазин, торгующий уникальной продукцией (*ассортимент кондитерских магазинов рассматривайте как данные перечислимого типа*)?
 8. Пусть заданы два предложения, слова в которых разделены запятыми или пробелами. Каждое предложение заканчивается точкой. Можно ли из букв первого предложения составить второе предложение и наоборот? Если нельзя ни то, ни другое, то перечислите буквы, которых не хватает в первом (*втором*) предложении, чтобы составить второе (*первое*).
 9. В столовой имеются отдельные меню на завтрак, обед и ужин. Известно, что в каждом из этих меню не более 100 видов блюд. Определите, какие виды блюд имеются и на завтрак, и на обед, и на ужин, если такие есть. Определите виды блюд, которые есть только на завтрак, только на обед, только на ужин (*виды блюд рассматривайте как данные перечислимого типа*).
 10. Вводится слово – образец. Затем вводится список слов (*не более 100*). Определите слова, в которых нет хотя бы одной буквы из слова-образца. Выведите также слова, а также буквы, которых нет в слове-образце.
 11. Многие игры дети начинают со считалок (*недлинный стихотворный текст*). Играющий, на которого попадет последнее слово текста, выходит из круга. Пусть в кругу стоит n детей. Текст считалки вводится с клавиатуры. Напечатайте номера детей в том порядке, в котором они выходят из круга.
 12. **Близнецы.** Два нечетных простых числа, разнящихся не два, называются близнецами. Например, 5 и 7, 11 и 13, 17 и 19, ... Напишите программу, печатающую все числа близнецы в интервале $[2..255]$. Для определения является ли число простым, используйте решето Эратосфена.
 13. С помощью решета Эратосфена найдите четверки, меньших n простых чисел, принадлежащих одному десятку (*например, 11, 13, 17, 19*).

Составители: Васильев Валерий Викторович,
Хливненко Любовь Владимировна

Редактор

Тихомирова О.А.